

## ON COMPUTING MENTAL STATES

*L. P. Ranatunga*

Department of Mathematics  
University of Brunei Darussalam  
Tel: 673-2-249001  
Fax: 673-2-249502  
email: lpr@ubd.edu.bn

### **ABSTRACT**

*An expert system has been created based on the Theravada Buddhist theory of the mind to compute coexisting mental states of ordinary human beings. Given one state of the mind, the system finds other coexisting set of states. The expert system supplies information about the constituents of consciousness. A stochastic model of impulsive thought moments in Buddhist psychology has been constructed.*

**Keywords:** *Buddhist psychology, Expert systems, PROLOG language, Stochastic model, Impulsive thought moments, Thought process, Neural networks.*

### **1.0 INTRODUCTION**

The Theravada Buddhist psychology deals with the mind, thought, thought-processes and mental states. From the Buddhist psychological standpoint no distinction is made between mind and consciousness. It should be pointed out that there are many technical terms in Buddhist psychology which cannot be rendered into English so as to convey their exact meaning. For the exact meaning of the technical terms used here, the reader is referred to several documents listed under References [2,4,5,6].

The analysis and classification of consciousness is given in Buddhist psychology. Based on the classification presented there, we start building up the knowledge base of an expert system. To begin with, we shall express each type of consciousness as a fact in PROLOG language. This will be the subject in section 2.

According to the Buddhist theory of mind, every consciousness is accompanied by a set of mental qualities which are called mental factors or states. The association of mental states to each type of consciousness can be expressed by adding relevant rules to our knowledge base. Section 3 deals with such rules.

Each consciousness performs at least one function. There are 14 functions performed by all of them. Section 4 deals

with such functions. Feeling is a mental state and there are five types of feeling which we experience in our day-to-day life. The rules for associating these five types of feeling to various types of consciousness will be dealt with in section 5. This completes our knowledge base. It is possible to extend this knowledge base to include more information, such as objects of various types of consciousness, however, no attempt has been made to do that here.

Having completed the knowledge base, it becomes possible to extract certain information that may be of psychological interest. The inference engine presented in section 6 performs this task. Impulsion is a function performed by certain types of consciousness. It is the phase of full cognition in the cognitive series or thought process occurring at its climax, if the respective object is large or distinct[4]. In a thought process there is a sequence of steps called thought moments or consciousness. The computation of sets of mental states associated with impulsive thought moments is included in section 6.

Computer modelling of thought process has been studied by Karunananda [1]. Their analysis concerns with the effect of determining consciousness on thought process. Considering three mental states, namely, attachment, aversion and equanimity, Karunananda calculates probabilities in future stages of the three mental states.

We shall build up a mathematical model to find out the behaviour of impulsive thought moments. This will be dealt with in section 7. The expert system supplies information about constituents of consciousness and this could be used to model thought process on a neural network. This is suggested as a further work in section 8.

### **2.0 CONSCIOUSNESS AND ITS REPRESENTATION AS FACTS**

The analysis and the classification of consciousness are given in Buddhist psychology. Consciousness is broadly classified under four divisions according to the planes in which it is experienced. The four divisions are:

- (a) consciousness pertaining to the sensuous universe,
- (b) consciousness pertaining to the universe of form
- (c) consciousness pertaining to the universe of formless entitles and
- (d) supramundane consciousness.

The sensuous universe is so called because sense-desires play predominant part in this universe. Those who develop ecstasies are born after death in the universe of form and universe of formless. In the universe of formless there is no body but only mind. Beings born in universe of form are supposed to possess very subtle material forms. The supramundane consciousness enables one to transcend this world of mind-body. Types of consciousness which are experienced by ordinary people are included only in the class (a). Thus, we shall consider only the class (a).

Consciousness pertaining to the sensuous universe is further divided into four sub-divisions namely, moral, immoral, karmically resultant and functional. Each of these sub-division is further divided so that collectively, there are 54 types of consciousness pertaining to the sensuous universe alone. As a rule, no two types of consciousness coexist and no consciousness arises without an object, either mental or physical.

Among the 54 types there are 9 types of functional consciousness which are not experienced by ordinary people so we shall not present them here. In the sequel, we shall explain each of the classes and express them as facts in Prolog<sup>1</sup>.

## 2.1 Moral Type Consciousness

There are 8 types of moral consciousness accompanied by pleasure or indifferent feeling associated with knowledge or not, spontaneous or not. These are connected with generosity and loving-kindness and hence they are with roots. This can be expressed as:

```

toc(m,ro,consc,s,pl,glk,aw).
toc(m,ro,consc,ns,pl,glk,aw).
toc(m,ro,consc,s,pl,glk,dw).
toc(m,ro,consc,ns,pl,glk,dw).
toc(m,ro,consc,s,ind,glk,aw).
toc(m,ro,consc,ns,ind,glk,aw).
toc(m,ro,consc,s,ind,glk,dw).
toc(m,ro,consc,ns,ind,glk,dw).

```

where m - moral type; ro - with roots; consc - consciousness; s - spontaneous and ns -non spontaneous; pl - pleasure and ind - indifference; glk - connected with generosity and loving-kindness; aw -

associated with knowledge and dw -dissociated with knowledge.

## 2.2 Immoral Type Consciousness

These types of consciousness are rooted either in attachment, hatred or delusion. Consciousness rooted in attachment may be accompanied either by pleasure or by indifference, connected with wrong views or not, spontaneous or not. Hence we get 8 types of consciousness rooted in attachment. This can be expressed in Prolog as follows:

```

toc(i,a,consc,s,pl,cwv,nil).
toc(i,a,consc,ns,pl,cwv,nil).
toc(i,a,consc,s,pl,dwv,nil).
toc(i,a,consc,ns,pl,dwv,nil).
toc(i,a,consc,s,ind,cwv,nil).
toc(i,a,consc,ns,ind,cwv,nil).
toc(i,a,consc,s,ind,dwv,nil).
toc(i,a,consc,ns,ind,dwv,nil).

```

where i - immoral type; a - rooted in attachment; cwv - connected with wrong view and dwv -disconnected with wrong view; nil - not applicable<sup>2</sup>.

Consciousness rooted in hatred is accompanied by displeasure and may be spontaneous or not. Hence we get two types of consciousness rooted in hatred which can be expressed as

```

toc(i,ill,consc,s,dp,cill,nil).
toc(i,ill,consc,ns,dp,cill,nil).

```

where ill - rooted in illwill; cill connected with illwill.

The consciousness rooted in delusion is accompanied by indifference, may be accompanied either by doubt or restlessness. Hence we get 2 types of consciousness rooted in delusion.

```

toc(i,ig,consc,nil,ind,cdoub,nil).
toc(i,ig,consc,nil,ind,crest,nil).

```

where ig - rooted in ignorance or delusion; cdoub - connected with doubts and crest - connected with restlessness.

## 2.3 Karmically Resultant Consciousness

Both moral and immoral consciousness constitute what, in Pali, are termed Kamma (action). Correctly speaking, kamma denotes the moral and immoral volitions and their concomitant mental factors, causing rebirth and shaping the destiny of beings. These karmical volitions become manifest as moral or immoral actions by body, speech and mind. Those types of consciousness that arise as the

<sup>1</sup>We shall be using Turbo Prolog.

<sup>2</sup> The use of turbo prolog forces us to keep the same seven tuple to represent consciousness.

inevitable results of these moral and immoral consciousness are called resultant. It should be understood that both kamma and resultant are purely mental. There are 23 types of resultant consciousness which are categorised into three classes, namely, moral resultant, immoral resultant and resultant.

### 2.3.1 Moral Resultant Consciousness Without Roots

There are 8 types of moral resultant consciousness which includes the 5 types of sensuous consciousness (i.e. seeing, hearing, etc.) with desirable objects. We express them by the following facts.

toc(mr,wr,eye\_consc,nil,ind,nil,nil).  
 toc(mr,wr,ear\_consc,nil,ind,nil,nil).  
 toc(mr,wr,nose\_consc,nil,ind,nil,nil).  
 toc(mr,wr,tongue\_consc,nil,ind,nil,nil).  
 toc(mr,wr,body\_consc,nil,pl,nil,nil).  
 toc(mr,wr,receiv\_consc,nil,ind,nil,nil).  
 toc(mr,wr,invest\_consc,nil,pl,nil,nil).  
 toc(mr,wr,invest\_consc,nil,ind,nil,nil).

where mr - moral resultant; wr - without roots; receiv\_consc - receiving consciousness and invest\_consc - investigating consciousness and eye\_consc- eye consciousness, etc.

### 2.3.2 Immoral Resultant Consciousness Without Roots

There are 7 types of immoral resultant consciousness which include the 5 types of sensuous consciousness without desirable objects. We express them by the following facts.

toc(ir,wr,eye\_consc,nil,ind,nil,nil).  
 toc(ir,wr,ear\_consc,nil,ind,nil,nil).  
 toc(ir,wr,nose\_consc,nil,ind,nil,nil).  
 toc(ir,wr,tongue\_consc,nil,ind,nil,nil).  
 toc(ir,wr,body\_consc,nil,pain,nil,nil).  
 toc(ir,wr,receiv\_consc,nil,ind,nil,nil).  
 toc(ir,wr,invest\_consc,nil,ind,nil,nil).

where ir - immoral resultant.

### 2.3.3 Resultant Consciousness

There are 8 types of resultant consciousness. These are identical with the types in section 2.1 except that these are karma-results. Thus one may have to replace letter m in 8 facts of section 2.1 with letter r which we shall use for resultant. As such we shall omit writing them here.

## 2.4 Functional Consciousness

There are 2 types of functional consciousness<sup>3</sup> which are experienced by ordinary people. The two types of functional consciousness could be expressed as:

toc(f,wr,fsda\_consc,nil,ind,nil,nil).  
 toc(f,wr,mda\_consc,nil,ind,nil,nil).

where f -functional, fsda\_consc - five sense-door adverting consciousness, mda\_consc - mind-door adverting consciousness.

## 3.0 MENTAL STATES

According to the Buddhist theory of mind every consciousness is accompanied by a set of mental qualities which are called mental factors or states. No consciousness exists without its mental states. Coexisting consciousness and mental states are related to one another by way of *association*. Thus, given a type of consciousness one may list out all the mental states associated with it.

According to the Buddhist psychology there are fifty two mental states. Among them, 41 are called constant mental states. The remaining 11 are inconstant mental states which arise occasionally and separately.

### 3.1 Constant Mental States and Association Rules

The constant mental states are:

contact, feeling, perception, volition, one\_pointedness, faculty\_of\_life, attention, initial\_application, sustained\_application, effort, decision, joy, conation, delusion, shamelessness, fearlessness, restlessness, attachment, misbelief, hatred, doubt, confidence, mindfulness, moral\_shame, moral\_dread, non\_attachment, goodwill, equanimity, tranquility\_of\_mental\_states, tranquility\_of\_mind, lightness\_of\_mental\_states, lightness\_of\_mind, pliancy\_of\_mental\_states, pliancy\_of\_mind, adaptability\_of\_mental\_states, adaptability\_of\_mind, proficiency\_of\_mind, proficiency\_of\_mental\_states, rectitude\_of\_mind, rectitude\_of\_mental\_states, wisdom.

We express the association of these mental states to various types of consciousness by the following rules in PROLOG.

universal([contact, feeling, perception, volition, one\_pointedness, faculty\_of\_life, attention],

---

<sup>3</sup> We have left out 9 other types as ordinary persons do not experience them.

```

toc(A,B,C,D,E,F,G)
  :- toc(A,B,C,D,E,F,G).
particulars([initial_application,      sustained_application,
effort], toc(A,B,consc,D,E,F,G))
  :- toc(A,B,consc,D,E,F,G).
particulars([initial_application,sustained_application,
decision], toc(A,B,receiv_consc,D,E,F,G))
  :- toc(A,B,receiv_consc,D,E,F,G).
particulars([initial_application,sustained_application,
decision], toc(A,B,invest_consc,D,E,F,G))
  :- toc(A,B,invest_consc,D,E,F,G).
particulars([initial_application,      sustained_application,
decision], toc(A,B,fsda_consc,D,E,F,G))
  :- toc(A,B,fsda_consc,D,E,F,G).
particulars([initial_application,      sustained_application,
decision, effort], toc(A,B,mda_consc,D,E,F,G))
  :- toc(A,B,mda_consc,D,E,F,G).
particulars([initial_application,      sustained_application,
decision, effort], toc(A,B,sp_consc,D,E,F,G))
  :- toc(A,B,sp_consc,D,E,F,G).
particulars([decision], toc(A,B,consc,D,E,F,G))
  :-toc(A,B,consc,D,E,F,G), F <>"cdoub".
particulars([joy], toc(A,B,C,D,pl,F,G))
  :-toc(A,B,C,D,pl,F,G), C <>"body_consc".
particulars([conation], toc(A,B,C,D,E,F,G))
  :-toc(A,B,C,D,E,F,G), B <>"wr",B <>"ig".
immoral([delusion,      shamelessness,      fearlessness,
restlessness], toc(i,B,C,D,E,F,G))
  :- toc(i,B,C,D,E,F,G).
immoral([attachment], toc(i,a,consc,D,E,F,G))
  :- toc(i,a,consc,D,E,F,G).
immoral([misbelief], toc(i,a,consc,D,E,cwv,nil))
  :- toc(i,a,consc,D,E,cwv,nil).
immoral([hatred], toc(i,ill,consc,D,dp,cill,nil))
  :- toc(i,ill,consc,D,dp,cill,nil).
immoral([doubt], toc(i,ig,consc,nil,ind,cdoub,nil))
  :- toc(i,ig,consc,nil,ind,cdoub,nil).
beautiful([confidence,      mindfulness,      moral_shame,
moral_dread, non_attachment, goodwill, equanimity,
tranquility_of_mental_states,      tranquility_of_mind,
lightness_of_mind,      lightness_of_mental_states,
pliancy_of_mind,      pliancy_of_mental_states,
adaptability_of_mind, adaptability_of_mental_states,
proficiency_of_mental_states,      proficiency_of_mind,
rectitude_of_mental_states, rectitude_of_mind],
toc(A,ro,consc,D,E,glk,G))
  :- toc(A,ro,consc,D,E,glk,G), A <>"i".
wish([wisdom], toc(A,ro,consc,D,E,glk,aw))
  :-toc(A,ro,consc,D,E,glk,aw), A <>"i".

```

### 3.2 Inconstant Mental States and Association Rules

The inconstant mental states are:

```

jealousy, avarice, worry, conceit, compassion,
appreciative_joy,      abstinence_from_wrong_speech,
abstinence_from_wrong_livelihood,
abstinence_from_wrong_action, sloth, torpor.

```

We express the association of these mental states to various types of consciousness by the following rules in PROLOG.

```

prob_i([conceit], toc(i,a,consc,D,E,dwv,nil))
  :- toc(i,a,consc,D,E,dwv,nil).
prob_i([jealousy],      toc(i,ill,consc,D,dp,cill,nil))
  :- toc(i,ill,consc,D,dp,cill,nil).
prob_i([avarice], toc(i,ill,consc,D,dp,cill,nil))
  :- toc(i,ill,consc,D,dp,cill,nil).
prob_i([worry], toc(i,ill,consc,D,dp,cill,nil))
  :- toc(i,ill,consc,D,dp,cill,nil).
prob_i([sloth],toc(i,B,consc,ns,E,F,nil))
  :- toc(i,B,consc,ns,E,F,nil).
prob_i([torpor], toc(i,B,consc,ns,E,F,nil))
  :- toc(i,B,consc,ns,E,F,nil).
abstinences([abstinence_from_wrong_speech],
toc(m,ro,consc,D,E,glk,G))
  :- toc(m,ro,consc,D,E,glk,G).
abstinences([abstinence_from_wrong_action],
toc(m,ro,consc,D,E,glk,G))
  :- toc(m,ro,consc,D,E,glk,G).
abstinences([abstinence_from_wrong_livelihood],
toc(m,ro,consc,D,E,glk,G))
  :- toc(m,ro,consc,D,E,glk,G).
illimitables([appreciative_joy], toc(m,ro,consc,D,pl,glk,G))
  :- toc(m,ro,consc,D,pl,glk,G).
illimitables([compassion], toc(m,ro,consc,D,pl,glk,G))
  :- toc(m,ro,consc,D,pl,glk,G).

```

We note that some of the above rules can be combined, however, as these mental states arise separately, it would be convenient to have a separate rule for each inconstant mental state for the purpose of computation.

## 4.0 RULES FOR FUNCTIONS

Each consciousness performs a particular function. Some types of consciousness perform several functions under different circumstances, in various capacities. Collectively, there are fourteen specific functions performed by them. They are; relinking, life-continuum, apprehending, seeing, hearing, smelling, tasting, contacting, receiving, investigating, determining, impulsion(Javana), retention and deacease. The following rules associate these functions to various types of consciousness.

```

functions([relinking,life_continuum, deacease],
toc(A,wr,invest_consc,nil,ind,nil,nil))
  :-toc(A,wr,invest_consc,nil,ind,nil,nil).

```

```

functions([relinking, life_continuum, decease, retention],
toc(r,ro,consc,D,E,glk,G))
:- toc(r,ro,consc,D,E,glk,G).
functions([apprehending],
toc(f,wr,fsda_consc,nil,ind,nil,nil))
:- toc(f,wr,fsda_consc,nil,ind,nil,nil).
functions([apprehending,determining],
toc(f,wr,mda_consc,nil,ind,nil,nil))
:-toc(f,wr,mda_consc,nil,ind,nil,nil).
functions([seeing,hearing,smelling,tasting, contacting,
receiving], toc(A,wr,receiv_consc,nil,ind,nil,nil))
:-toc(A,wr,receiv_consc,nil,ind,nil,nil).
functions([investigating, retention],
toc(A,wr,invest_consc,nil,E,nil,nil))
:-toc(A,wr,invest_consc,nil,E,nil,nil).
functions([impulsion], toc(i,B,consc,D,E,F,nil))
:- toc(i,B,consc,D,E,F,nil).
functions([impulsion], toc(m,ro,consc,D,E,glk,G))
:- toc(m,ro,consc,D,E,glk,G).

```

## 5.0 SUMMARY OF FEELING

Feeling is a significant mental state which is common to all types of consciousness. Feeling is fivefold, namely, happiness, pain, pleasure, displeasure, and indifference. In this section, we shall present rules for associating them to various types of consciousness.

```

accompany(happiness,
toc(mr,wr,body_consc,nil,pl,nil,nil))
:- toc(mr,wr,body_consc,nil,pl,nil,nil).
accompany(pain, toc(ir,wr,body_consc,nil,pain,nil,nil))
:- toc(ir,wr,body_consc,nil,pain,nil,nil).
accompany(pleasure,toc(i,a,consc,D,pl,F,nil))
:- toc(i,a,consc,D,pl,F,nil).
accompany(pleasure, toc(A,ro,consc,D,pl,glk,G))
:-toc(A,ro,consc,D,pl,glk,G).
accompany(pleasure, toc(mr,wr,invest_consc,nil,pl,nil,nil))
:- toc(mr,wr,invest_consc,nil,pl,nil,nil).
accompany(pleasure, toc(f,wr,ppl_consc,nil,pl,nil,nil))
:- toc(f,wr,ppl_consc,nil,pl,nil,nil).
accompany(displeasure, toc(i,ill,consc,D,dp,cill,nil))
:- toc(i,ill,consc,D,dp,cill,nil).
accompany(indifference, toc(A,ro,consc,D,ind,glk,G))
:-toc(A,ro,consc,D,ind,glk,G).
accompany(indifference, toc(A,wr,C,nil,ind,nil,nil))
:-toc(A,wr,C,nil,ind,nil,nil).
accompany(indifference, toc(i,B,consc,D,ind,F,nil))
:-toc(i,B,consc,D,ind,F,nil).

```

## 6.0 INFERENCE ENGINE

In sections 2 to 5, we have constructed our knowledge base. In this section, we shall construct the inference engine which will allow us to extract some information from our knowledge base.

### 6.1 Computing Constant Mental States

To compute the set of coexisting constant mental states of a given type of consciousness the following two rules can be used.

```

fixed(S,toc(A,B,C,D,E,F,G))
:- universal(S,toc(A,B,C,D,E,F,G));
particulars(S,toc(A,B,C,D,E,F,G));
immoral(S,toc(A,B,C,D,E,F,G));
beautiful(S,toc(A,B,C,D,E,F,G));
wish(S,toc(A,B,C,D,E,F,G)).
f_states(L,toc(A,B,C,D,E,F,G))
:-instantiate(toc(A,B,C,D,E,F,G)),
findall(S, fixed(S,toc(A,B,C,D,E,F,G)),L).

```

Where in the last rule, the predicate instantiate is given by  
instantiate(T) :- T.

and we have used Turbo Prolog findall predicate to collect backtracking values into the list L. It is worth pointing out that the goal `f_states(L,T)` will succeed for any value of T as there are 7 mental states common to each type of consciousness (see the universal predicate in section 3.1).

### 6.2 Computing Inconstant Mental States

The inconstant mental states arise separately as such there can be only one at a time. To compute an inconstant mental state of a given type of consciousness the following rule can be used.

```

unfixed(S,toc(A,B,C,D,E,F,G))
:- prob_i(S,toc(A,B,C,D,E,F,G));
abstinences(S,toc(A,B,C,D,E,F,G));
illimitables(S,toc(A,B,C,D,E,F,G)).

```

Notice that the goal `unfixed(S,T)` will not succeed for each value of T as some types of consciousness has no inconstant mental states.

### 6.3 Computing Coexisting Mental States of A Given Type of Consciousness

If T is a variable representing type of consciousness and L is a variable to hold set of mental states then coexisting mental states of a given type of consciousness can be computed by the following rules.

```

msconsc2(L,T) :- f_states(SS,T), makelist(SS,L1),
unfixed(L2,T), append(L1,L2,L).

```

msconsc2(L,T) :- f\_states(SS,T), makelist(SS,L).

The makelist(L1,L2) predicate will construct a single list of states L2 from the list of list of states. The predicate append(L1,L2,L) will append the list L2 to end of L1 and places the result in L.

We note that the inconstant mental states are present occasionally as such we need two rules for msconsc2.

#### 6.4 Computing Coexisting Mental States for All Types of Consciousness

We shall use msconsc2 and findall predicates to collect all backtracking values to list LL as in

msconsc1(LL) :- findall(L,msconsc2(L,T),LL).

The list of lists LL thus obtained may contain copies of the same list L. To remove them a predicate r\_dup is introduced. r\_dup(L1,L2) will construct a list of lists L2 by removing copies in the list of lists L1. Now, to find coexisting mental states of all types of consciousness we can write a rule as in

msconsc(L):- msconsc1(LL), r\_dup(LL,L).

The goal msconsc(L) results in L containing 46 lists. Since we have 45 types of consciousness getting 46 lists may require some explanation. It was mentioned that inconstant mental states arise separately and occasionally. This means that when more than one inconstant state are associated to a type of consciousness it generates more than one list. For example sloth and torpor are present in non-spontaneous, immoral types of consciousness. Thus each such type of consciousness generates three lists: one with torpor, one with sloth and one without both of them. Two types of consciousness do not necessarily differ in terms of mental states. Two types may differ in terms of its states, functions, energies, or aspects, or whatever other name we choose to give to the "components" of consciousness.

We shall list below all the possible sets of coexisting mental states for 46 types of consciousness.

- (1) contact, feeling, perception, volition, one\_pointedness, faculty\_of\_life, attention.  
Length of the list is 7
- (2) contact, feeling, perception, volition, one\_pointedness, faculty\_of\_life, attention, initial\_application, sustained\_application, decision.  
Length of the list is 10
- (3) Set of states in (2) and effort.  
Length of the list is 11
- (4) Set of states in (2) and joy.  
Length of the list is 11

- (5) contact, feeling, perception, volition, one\_pointedness, faculty\_of\_life, attention, initial\_application, sustained\_application, effort, decision, conation, confidence, mindfulness, moral\_shame, moral\_dread, non\_attachment, goodwill, equanimity, tranquility\_of\_mental\_states, tranquility\_of\_mind, lightness\_of\_mental\_states, lightness\_of\_mind, pliancy\_of\_mental\_states, pliancy\_of\_mind, adaptability\_of\_mental\_states, adaptability\_of\_mind, proficiency\_of\_mental\_states, proficiency\_of\_mind, rectitude\_of\_mental\_states, rectitude\_of\_mind.  
Length of the list is 31
- (6) Set of states in (5) and Joy.  
Length of the list is 32
- (7) Set of states in (5) and abstinence\_from\_wrong\_livelihood.  
Length of the list is 32
- (8) Set of states in (5) and abstinence\_from\_wrong\_action.  
Length of the list is 32
- (9) Set of states in (5) and abstinence\_from\_wrong\_speech,  
Length of the list is 32
- (10) Set of states in (5), joy and compassion.  
Length of the list is 33
- (11) Set of states in (5), joy and appreciative\_joy  
Length of the list is 33
- (12) Set of states in (5), joy and abstinence\_from\_wrong\_livelihood.  
Length of the list is 33
- (13) Set of states in (5), joy and abstinence\_from\_wrong\_action.  
Length of the list is 33
- (14) Set of states in (5), joy and abstinence\_from\_wrong\_speech.  
Length of the list is 33
- (15) Set of states in (5) and wisdom.  
Length of the list is 32
- (16) Set of states in (5), joy and wisdom.  
Length of the list is 33
- (17) Set of states in (5), wisdom and abstinence\_from\_wrong\_livelihood.  
Length of the list is 33
- (18) Set of states in (5), wisdom and abstinence\_from\_wrong\_action.  
Length of the list is 33
- (19) Set of states in (5), wisdom and abstinence\_from\_wrong\_speech  
Length of the list is 33
- (20) Set of states in (5), joy, wisdom, compassion,  
Length of the list is 34
- (21) Set of states in (5), joy, wisdom and appreciative\_joy.  
Length of the list is 34

- (22) Set of states in (5), joy, wisdom and abstinence\_from\_wrong\_livelihood.  
Length of the list is 34
- (23) Set of states in (5), joy, wisdom and abstinence\_from\_wrong\_action.  
Length of the list is 34
- (24) Set of states in (5), joy, wisdom and abstinence\_from\_wrong\_speech.  
Length of the list is 34
- (25) contact, feeling, perception, volition, one\_pointedness, faculty\_of\_life, attention, initial\_application, sustained\_application, decision, effort, delusion, shamelessness, fearlessness, restlessness.  
Length of the list is 15
- (26) contact, feeling, perception, volition, one\_pointedness, faculty\_of\_life, attention, initial\_application, sustained\_application, effort, delusion, shamelessness, fearlessness, restlessness, doubt.  
Length of the list is 15
- (27) Set of states in (25), hatred and conation.  
Length of the list is 17
- (28) Set of states in (25), conation and attachment.  
Length of the list is 17
- (29) Set of states in (25), conation, attachment, and misbelief.  
Length of the list is 18
- (30) Set of states in (25), joy, conation and attachment.  
Length of the list is 18
- (31) Set of states in (25), joy, conation, attachment and misbelief.  
Length of the list is 19
- (32) Set of states in (25), conation, hatred, and worry.  
Length of the list is 18
- (33) Set of states in (25), conation, hatred and avarice.  
Length of the list is 18
- (34) Set of states in (25), conation, hatred and jealousy.  
Length of the list is 18
- (35) Set of states in (25), conation, attachment and conceit.  
Length of the list is 18
- (36) Set of states in (25), joy, conation, attachment and conceit.  
Length of the list is 19
- (37) Set of states in (25), conation, hatred and torpor.  
Length of the list is 18
- (38) Set of states in (25), conation, hatred and sloth.  
Length of the list is 18
- (39) Set of states in (25), conation, attachment and torpor.  
Length of the list is 18
- (40) Set of states in (25), conation, attachment and sloth.  
Length of the list is 18
- (41) Set of states in (25), conation, attachment, misbelief and torpor.

- Length of the list is 19
- (42) Set of states in (25), conation, attachment, misbelief and sloth.  
Length of the list is 19
- (43) Set of states in (25), joy, conation, attachment and torpor.  
Length of the list is 19
- (44) Set of states in (25), joy, conation, attachment and sloth.  
Length of the list is 19
- (45) Set of states in (25), joy, conation, attachment, misbelief and torpor.  
Length of the list is 20
- (46) Set of states in (25), joy, conation, attachment, misbelief and sloth.  
Length of the list is 20

### 6.5 Computing Coexisting Mental States of A Given Mental State

Psychologists may be interested to find out what are the coexisting mental states of a given mental state. For example, what are the other coexisting mental states when we are angry. To find this we may write rules as in

```

coexist(X,L)      :- coexist1(X,LL), r_dup(LL,L).
coexist1(X,LL)   :- findall(L, coexist2(X,T,L), LL).
coexist2(X,T, L) :- fixed(S,T), member(X,S),
                  f_states(S1,T), makelist(S1,L1),
                  unfixed(S2,T), append(L1,S2,L).
coexist2(X,T, L) :- fixed(S,T),member(X,S),
                  f_states(S1,T), makelist(S1,L).
coexist2(X,T, L) :- unfixed(S,T), member(X,S),
                  f_states(S2,T), makelist(S2,L2), append(L2,S,L).
    
```

In predicate coexist (X,L), X is the variable representing the given mental state and L is the variable which holds the list of possible mental state sets required. In coexist2, T is a variable representing the type of consciousness and hence for each possible value of T, coexist2 will find a set of coexisting mental states of X and the coexist1 will collect all the backtracking values to LL.

The goal coexist(hatred, L) results in L containing the lists (27), (32), (33), (34), (37) and (38) of section 6.4.

### 6.6 Computing Mental States Associated With Functions

In section 4, we presented rules for associating fourteen kinds of functions to various types of consciousness. As one function is performed by more than one type of consciousness, there may be more than one set of coexisting mental states associated with each function. To compute these sets of mental states the following rules may be used.

```

fun_states(Name,SS)
  :- findall(S,fun_states1(Name,T,S),S1), r_dup(S1,SS).
fun_states1(Name,T,SS)
  :- functions(S,T), member(Name, S), f_states(S1,T),
  makelist(S1,L1), unfixed(S2,T), append(L1,S2,SS).
fun_states1(Name,T,SS)
  :- functions(S,T), member(Name, S), f_states(S1,T),
  makelist(S1,SS).

```

The predicate `fun_states1` takes three arguments, the first variable, `Name`, takes a symbol or atom as its value and the second variable, `T`, takes a type of consciousness as its value and the third variable, `SS`, then holds the set of mental states. The predicate `fun_states` simply collects all the backtracking values and then removes copies of the same list entries.

The goal `fun_states(impulsion, L)`, for example, results in `L` containing 42 lists. They are the lists from (5) to (46) given in section 6.4.

## 6.7 Computing Coexisting Mental States of A Type of Feeling

In section 5, we associated 5 types of feeling with types of consciousness. We can make use of those knowledge-based rules to compute coexisting mental states of each type of feeling.

```

feel_states(L)
  :-findall(S,feel_states1(Name,T,S),SS), r_dup(SS,L).
feel_states1(Name, T, L)
  :-accompany(Name, T), f_states(S1,T),
  makelist(S1,L1), unfixed(L2,T), append(L1,L2,L).
feel_states1(Name, T, L)
  :-accompany(Name, T), f_states(S1,T), makelist(S1,L).

```

The predicate `feel_states1(Name, T, L)` takes three variables. The first variable, `Name`, can hold the symbol for given type of feeling, the second variable, `T` takes its value as a type of consciousness and the third variable, `L` can hold the coexisting mental states for the type of feeling specified by `Name`. The predicate `feel_states(L)` we collect all the backtracking values of the third variable of `feel_states1`, remove copies of the same list entries and put the result in `L`.

## 7.0 MODELLING THE IMPULSIVE THOUGHT MOMENTS

In this section, we shall construct a mathematical model based on probability theory to find the behaviour of impulsive thought moments. The model is applicable only for general human beings.

## 7.1 Thought Process

According to the Buddhist theory of mind every being thinks in accordance with a special mechanism known as the thought process. This mechanism is common to all beings and is a universal law. There are two types of thought processes. The first type, known as five-sense-door thought process, will receive input information from the sensory system. The second type, known as mind-door thought process will receive input information from the mind itself.

In a thought process, there is a sequence of steps called thought moments or consciousness. Each step has some sort of function in the process. There are 17 thought moments in the sense-door-thought process. Both thought processes, include a determining state. Discrimination is exercised at this stage. Freewill plays its part here. This is the gateway to a moral or immoral thought process. In both thought processes, immediately after the determining state comes the psychologically most important stage impulsion or *Javana*. It is at this stage that an action is judged whether moral or immoral. If viewed rightly, at the determining stage, the impulsion become moral; if viewed wrongly it becomes immoral. There are normally seven impulsive thought moments. The mental states occurring in all these thought moments are similar, but their intensity differs.

The initial state of both the processes is the life-continuum consciousness. Its function is to keep the stream flowing. The flow of life-continuum is interrupted when objects enter the mind. Owing to the rapidity of the flow of life-continuum an external object does not immediately give rise to a thought-process. If the process is created, then as a rule, for a complete perception of a physical object through one of the sense-doors precisely 17 thought moments must pass and then subside to life-continuum.

When an object is presented to our minds through our sensory system after the five-sense-door thought process there arises a mind-door thought process perceiving the object mentally. Again the stream of consciousness subsides into life-continuum and two more similar thought processes arise before the object is actually known. As each of the thought processes normally includes the phase of impulsion there arise at most four phases of impulsion before the object is actually known.

We shall build up a stochastic model to calculate the probability of impulsion becoming moral. According to the Buddhist theory of mind, there are three moral roots; wisdom, non-attachment and goodwill. There are three immoral roots; delusion, attachment and hatred. It should be pointed out that a mental phenomenon is called 'moral'



if actually associated with a moral root; 'immoral', if associated with an immoral root. Let us suppose that an object enters the mind and it reaches impulsive thought moments. Thought moments with moral and immoral root-conditions cannot follow each other immediately, but appear in different serial process of cognition, each of which has to start anew with the phase of life-continuum[5]. Thus, in a single thought process when the initial impulsive thought moment becomes moral the subsequent impulsive thought moments are moral for that particular thought process.

In the sequel, we shall use IBM to denote the event 'impulsion becoming moral' and IBI to denote the event 'impulsion becoming immoral'. We write P for the probability function. It should be pointed out that IBI and IBM are mutually exclusive and exhaustive events for ordinary people and hence, we can compute P(IBM), using the formula  
 $P(IBM) = 1 - P(IBM)$ .

### 7.2 Transitions in Impulsion

It was mentioned earlier that when an object enters the mind there arise at most four phases of impulsion before the object is actually known.

Now, suppose, at the initial phase, impulsion becomes moral. Since there are at most three more phases of impulsion before the object is actually known, subsequent phases of impulsion could be represented by the following diagram




Fig. 1: Phases of impulsion

Where M - moral and I -immoral. We have left out in between thought moments such as life-continuum from the above diagram.

Let us assume that the probability of next impulsion depends only on the current impulsion. Under this assumption, the above phases of impulsion represent two state Markov chain[3]. Although we should consider only four phases of impulsion, we shall construct a model in general so that probability for any phase could be

computed. The idea behind this is to compare the probability value at the 4<sup>th</sup> phase of impulsion with the probability of limiting state[3] of the Markov chain, if exists.

Let us identify the two states as 0(moral), 1(immoral) and represent the Markov chain by the random variables of the family  $\{X_n, n=0, 1, 2, \dots\}$ .

Let the one-step transition probability be defined by  
 $P_{ij} = P(X_{n+1} = j | X_n = i)$ .

This is the conditional probability of state j, given state i. Then, the transition probabilities from state i to j can be represented as a matrix T, which is called a probability transition matrix[3]. For our model this probability transition matrix can be written as

$$T = \begin{bmatrix} P_{00} & P_{01} \\ P_{10} & P_{11} \end{bmatrix}$$

Let  $P_j^{(n)}$  be the unconditional distribution of the state of the Markov chain after n steps defined by

$$P_j^{(n)} = P(X_n = j)$$

Then

$$P_0^{(n)} = P_0^{(n-1)}P_{00} + P_1^{(n-1)}P_{10} \quad (1)$$

For any phase of impulsion, the sum of the probabilities of impulsion becomes moral and impulsion become immoral must be equal to one. We therefore have,

$$P_0^{(n-1)} + P_1^{(n-1)} = 1 \quad (2)$$

The sum of transition probabilities from state 0 to 0 and from 0 to 1 must be equal to 1.

Thus, we have

$$P_{00} = 1 - P_{01} \quad (3)$$

From equations (1), (2) and (3) we obtain

$$P_0^{(n)} = P_0^{(n-1)}(1 - P_{01} - P_{10}) + P_{10}$$

This is a non-homogeneous recurrence relation with constant coefficients. Let  $P_{IBM}$  be a particular value of the probability of the initial impulsion become moral. Let us use  $P_0^0 = P_{IBM}$  as the initial condition. Solving the above recurrence relation subject to the initial condition yields

$$P_0^{(n)} = (P_{IBM} - \frac{P_{10}}{P_{01} + P_{10}})(1 - P_{01} - P_{10})^n + \frac{P_{10}}{P_{01} + P_{10}}$$

for all  $n \geq 0$ , provided that  $P_{01} + P_{10} \neq 0$ .

We observe that when  $|1 - P_{01} + P_{10}| < 1$

$$\lim_{n \rightarrow \infty} P_0^{(n)} = \frac{P_{10}}{P_{01} + P_{10}}$$

which we call the limiting probability of impulsion become moral. Let  $r$  be this limit. It would be interesting to compare the value of  $r$  with the value of our model at the 4<sup>th</sup> phase of impulsion (i.e. the value of our model when  $n = 3$ ).

### 7.3 Basis for Assigning Probabilities to Transition Matrix T

The probability of the same event occurring in the next stage is greater than that of a different event occurring in the next stage.

We can summarise this condition as  $P_{00} > P_{01}$  and  $P_{11} > P_{10}$ .

Now, since  $P_{00} + P_{01} = 1$  and  $P_{11} + P_{10} = 1$ , these two inequalities result

$$P_{01} < 0.5 \text{ and } P_{10} < 0.5.$$

We shall tabulate some probabilities of our model. We set  $P_{IBM} = 20/42$  (see comments below). The table 1 compares values of the limiting probability with the values of our model when impulsion reaches its fourth phase.

Table 1: Limiting probabilities

$P_{01}$	$P_{10}$	$P_0^{(3)}$	$r$
0.3	0.3	0.4985	0.5000
0.2	0.3	0.5845	0.6000
0.2	0.1	0.3823	0.3333
0.35	0.30	0.4622	0.4615
0.40	0.40	0.4998	0.5000
0.49	0.49	0.5000	0.5000

It can be seen that  $r$  and  $P_0^{(3)}$  are approximately the same for many values of our probability selections.

### 7.3.1 Comments

We first discuss why the value 20/42 was chosen for  $P_{IBM}$ . In section 6.6, we found that the sets of coexisting mental states associated with impulsive consciousness to be the sets numbered by 5 to 46. Thus, we have 42 possible existence of impulsive thought moments. We shall assume their existence as equally likely. Among them at least one of the moral roots (wisdom, goodwill, non attachment) is present in 5 to 24. Therefore, it is reasonable to set  $P_{IBM} = 20/42$ .

It was mentioned earlier that when the impulsion runs for 5 or 7 consecutive moments the intensity of mental factors differ in impulsive thought moments. It may be this difference in intensity of mental factors which results in different values of  $P_0^{(n)}$  as  $n$  increases.

## 8.0 CONCLUSION AND SUGGESTIONS FOR FURTHER WORK

An expert system has been created to compute mental states. The system allows one to extract information about other coexisting mental states given one component of the set of mental states.

Under the assumption that next state of impulsion depends only on current state, a stochastic model of impulsive thought moments has been built. The model allows one to compute the probability of impulsion become moral or immoral for any moment.

The expert system could be used to compute mental states associated with thought moments. This information about constituents of consciousness could be used to model thought processes on a neural network. One may use populations of neurons to model mental states. Co-existing mental states can then be represented by a layer of relevant neuron populations. Characteristics of the mental states are obtainable [2,4,5,6]. This additional information about mental states could be used to connect one layer of neuron to the other. Because of the cyclic nature of the thought processes (starts and ends at life-continuum) we may expect network to exhibit multi-directional associative memory. Initial work has been carried out in this direction.

### ACKNOWLEDGEMENTS

The author would like to thank referees of this paper for their valuable comments.

## REFERENCES

- [1] A. S. Karunananda, "Computer modelling of thought processes", *International Journal of Computer Applications in Technology*, Vol. 6, No. 2/3, 1993.
- [2] Narada Maha Thera, *A Manual of Abhidhamma*, Fifth Edition, 1987, Buddhist Missionary Society, Malaysia.
- [3] U. Narayan Bhat, *Elements of Applied Stochastic Processes*, Second Edition, 1984, John Wiley & Sons.
- [4] Nyanatiloka, *Buddhist Dictionary*, Fourth revised edition, 1980, Buddhist publication society, Sri Lanka.
- [5] Nyanatiloka, *Guide through the Abhidhamma-Pitaka*, Fourth edition, 1983, Buddhist publication society, Sri Lanka.
- [6] C. A. F. Rhys Davids, *A Buddhist Manual of Psychological Ethics*, Third edition, 1993, The Pali text society, Oxford.

## BIOGRAPHY

L P Ranatunga obtained his Ph.D. in Computer Science from University of Kent at Canterbury in 1989. Currently, he is working as a lecturer at the Faculty of Science, University of Brunei Darussalam. His research areas include concurrency theory and expert systems.